# Maintainer: tobias <tobias@archlinux.org>

pkgname=('gconf-sharp')
pkgver=2.24.2
pkgrel=6
arch=(x86_64)
license=(LGPL)
url="http://gtk-sharp.sourceforge.net"
pkgdesc="Mono bindings for GConf"
depends=('gtk-sharp-2' 'gconf')
source=(https://download.gnome.org/sources/gnome-sharp/2.24/gnome-sharp-${pkgver}.tar.bz2
        02_fix_pkg-config_paths.patch
        04_initialize_dbus_glib_threading.patch)
md5sums=('3b38f53960c736d4afb8f04204efe98b'
         'a8c1bf57a384f5fc20a5890f9f1cef5a'
         '90cd7ea88aaa3011522376153433d115')

prepare() {
  cd gnome-sharp-${pkgver}
  patch -Np1 -i ../04_initialize_dbus_glib_threading.patch
  patch -Np1 -i ../02_fix_pkg-config_paths.patch
}

build() {
  cd gnome-sharp-${pkgver}
  ./configure --prefix=/usr --sysconfdir=/etc
  make
}

package() {
  cd gnome-sharp-${pkgver}
  make -C gconf/GConf install DESTDIR="${pkgdir}"
  make -C gconf/tools install DESTDIR="${pkgdir}"
}
